Chapter 17
Geotechnical Module

CivilFEM Manual of Essential Examples
Chapter 17 – Table of Contents

Example 17.1
Earth Pressure on Shell Elements

Example Description

The cross section of a retaining wall is shown in the figure bellow. It is designed as a simple beam in flexure. The height of the wall is variable. Use shell type elements and the provided information to determine the earth pressure distribution and the reinforcement of this retaining wall subjected to a uniform distributed surcharge q.



Figure 17.1-1 Problem Sketch

Geometric Properties

a = 8 m

b = 6 m

c = 10 m

d = 0.6 m

Loads

q = 2 KN/m

Materials (Eurocode 2)

Concrete: C20/25 ; γ_c = 1.5

Steel: S-400 ; γ_s = 1.15

Soil Characteristics

Compact sand

γ = 21 KN/m^3

δ = 0º

φ = 30º

n = 25%

C = 0

Analysis, Assumptions and Modeling Notes

Meters and Kilonewton are used as length and stress units. Shell63 elements have been used for this model.



Figure 17.1-2 Keypoints Location

Input Data Listing

```
 FINISH
~CFCLEAR,,1  ! Not needed

! CivilFEM Setup: Code & Units
 ~UNITS,,LENG,M                     ! m, sg, Kn
 ~UNITS,,TIME,S
 ~UNITS,,FORC,KN
 ~CODESEL,EC3-05,EC2-08,,,EC8-04    ! Eurocodes 2, 3 and 8

/PREP7
! CivilFEM Preprocessor
! --------------------------------------------------
! Material: Concrete C20/25 (gc=1.5) and steel S400 (gs=1.15)
  ~CFMP,1,LIB,CONCRETE,EC2,C20/25,0,0,0
```

```
  ~CFMP,2,LIB,REINF,EC2,S400,0,0,0
  ~CFMP,3,LIB,SOIL,,SW                  ! Sand well granulated

! Modify soil properties
! Angle of internal friction  PHI = 30
  ~CFMP,3,SOIL    ,GAMD ,,21          ! Specific weight of soil
  ~CFMP,3,SOIL    ,N    ,,0.25        ! Index of holes

! Terrain definition
! Soil surface level: 8
! Slope of earth:     0
  ~TERDEF,1,NEW,1,0,Y,8,0,2,0,,
  ~TERDEF,1,LAYER,1,3,8, ,CLMB,

! Element type: 4 Nodes Shell
  ET,1,SHELL63

! Wall section (Thickness = 0.60 m; cover = 5 cm)
  ~SHLRNF,1,0.6000,1,2,0.05,0.0,0.0,0.0,0.0,0,0.0,45.0
  ~BMSHPRO,1,SHELL,1,1,1,1,63,,, ! Wall

! Ansys Preprocessor
! --------------------------------------------------
! Solid model
  K,1,0,0
  K,2,10,2
  K,3,10,8
  K,4,0,8
  A,1,2,3,4
! Mesh
  ESIZE,1  ! Size of elements = 1.00 m
  MSHKEY,1 ! Use only quads
   ! Reinforcement coordinate system
    LOCAL,11,0 ! Coordinate System parallel to the global system
    ESYS,11    ! Assign esys 11 to elements
   ! Mesh
    AMESH,1
! Plot model
  /VIEW,,1,1,1  ! Iop view
  /ESHAPE,1
  EPLOT

! Boundary Conditions
  LSEL,S,,,1
  NSLL,S,1
  D,ALL,ALL
  NSEL,ALL

! Apply ACTIVE earth pressures
  ~ETHSFE,2,ACT,1,,,
! Plot of earth pressures
  /PSF,PRES,2
  EPLOT

/SOLU
! Ansys Solution
! --------------------------------------------------
  SOLVE

/POST1
~CFSET,,1
! CivilFEM Postprocessor
! --------------------------------------------------
! Design shell reinforcements with C20/25 and S400
   ~DIMCON,SHELL,WOOD
! Plot results
   ! Horizontal external (Bot) reinforcement area/m
    ~PLSHCON,ASBX
```

```
! horizontal interior (Top) reinforcement area/m
  ~PLSHCON,ASTX
! vertical external (Bot) reinforcement area/m
  ~PLSHCON,ASBY
! vertical interior (Top) reinforcement area/m
  ~PLSHCON,ASTY
```

CivilFEM Results



Figure 17.1-3 Earth Pressure Distribution



Figure 17.1-4 Reinforcement Required in X Direction at Bottom



Figure 17.1-5 Reinforcement Required in X Direction at Top



Figure 17.1-6 Reinforcement Required in Y Direction at Bottom



Figure 17.1-7 Reinforcement Required in Y Direction at Top

Example 17.2
Earth Pressure on Solid Elements

Example Description

The cross section of a trapezoidal retaining wall is shown in the bellow figure. Calculate the factor of safety with respect to overturning and the earth pressure diagram. Use solid type elements and the provided information.

Geometric Properties

a = 5 m

b = 0.5 m

c = 2.5 m



Figure 17.2-1 Problem Sketch

Materials (Eurocode 2)

Concrete: C16/20

ρ = 24 KN/m^3

Soil Characteristics

Granular Filled

γ = 18 KN/m^3

δ = 20º

φ = 30º

C = 0

Analysis, Assumptions and Modeling Notes

Meters and Kilonewton are used as length and stress units. Plane strain PLANE42 elements have been used for this model.

Input Data Listing

```
 FINISH
~CFCLEAR,,1! Not needed

! CivilFEM Setup: Code & Units
 ~UNITS,,LENG,M                 ! m, sg, Kn
 ~UNITS,,TIME,S
 ~UNITS,,FORC,KN
  ~CODESEL,EC3-05,EC2-08,,,EC8-04         ! Eurocodes No.2, No.3 and No.8

/PREP7
! CivilFEM Preprocessor
! --------------------------------------------------
! Material: Concrete C16/20, Specific Weigh=24 KN/m3
  ~CFMP,1,LIB,CONCRETE,EC2,C16/20
  ~CFMP,1,USER
  ~CFMP,1,DatGen,GAM,,24
  ~CFMP,3,LIB,SOIL,,SW           ! Sand well granulated

! Modify soil properties
! Angle of internal friction  PHI = 30
  ~CFMP,3,SOIL,GAMD,,18,0,0,0   !Specific weight of soil

! Terrain definition
! Soil surface level: 5
! Slope of earth:     0
! Angle of terrain-wall friction: DELTA=20
 ~TERDEF,1,NEW,1,0,Y,5,0,0,0,,
 ~TERDEF,1,LAYER,1,3,5, ,CLMB,20

! Ansys Preprocessor
! --------------------------------------------------
! Element type 1: Solid-2D
  ET,1,PLANE42
  KEYOPT,1,3,2 ! Plane strain
! Solid model
  K,1,0,0
  K,2,2.5,0
  K,3,0.5,5
  K,4,0,5
! Area
  A,1,2,3,4
! Mesh
  ESIZE,0.5  ! Size of the elements
  MSHKEY,1   ! Use only quads
  AMESH,1    ! Mesh
! Boundary Conditions
  NSEL,S,LOC,Y,0
  D,ALL,ALL,0
  NSEL,ALL

! CivilFEM Earth Pressure
```

```
! --------------------------------------------------
! Select Nodes that define the pressure plane
! and elements on that plane to which the
! pressure will be applied
  LSEL,S,,,2
  NSLL,S,1
  ESLN,S

! Apply active earth pressures
  ~ETHSF,ACT,1,,,

! Plot of earth pressure
  /PSF,PRES,NORM,2
  /PBC,U,,1
  /PBC,F,,1
   ALLSEL,all
   EPLOT

/SOLU
! Ansys Solution

! --------------------------------------------------
! Self weight
  ACEL,,9.81
! Solve
  SOLVE
! Delete self weight
  ACEL,

! Ansys Solution
! --------------------------------------------------
! Solve
  SOLVE
! Delete loads (Not needed)
  SFEDELE,ALL,ALL,ALL
  FDEL,ALL

! Ansys Postprocessor
! --------------------------------------------------
! Enter the postprocessor module
  /POST1
! Sum forces in the wall base
  SPOINT,,0,0
! First load state
  SET,1
! Sum forces (Gravity=> Resisting moment)
  FSUM
! Get Stabilizer moment
  *GET,ME,FSUM,0,ITEM,MZ
! Second load state => Overturning moment
  SET,2
! Sum forces
  FSUM
! Get Overturning moment
  *GET,MD,FSUM,0,ITEM,MZ
! Safety factor against overturning
  CS=-ME/MD
! List Safety factor against overturning
  *STAT,CS
```

CivilFEM Results



Figure 17.2-2 Earth Pressure Distribution

```
PARAMETER STATUS- CS   (    7 PARAMETERS DEFINED)
                  (INCLUDING      4 INTERNAL PARAMETERS)

 NAME                              VALUE                        TYPE  DIMENSIONS
 CS                              0.688246899                     SCALAR
```

Example 17.3
Slope Stability Analysis

Example Description

The figure below presents a schematic diagram of a slope stability problem. The objective is to compute the minimum factor of safety and locate the critical slip surface location using both Fellenius and Bishop Method.

The slope is formed by 4 different layers which material properties are defined bellow. The contouring keypoints coordinates define the geometric properties of each layer. The piezometric line depicts the pore-water pressure conditions. A seismic load is applied in means of a seismic horizontal coefficient defined as 0.3 and a concentrated load is applied on the middle of the slope.

Special Features: Circular slip surfaces, 30 points in search grid, multiple soil layers, pore-water pressure specified by a piezometric line and a seismic load.





Figure 17.3-1 Problem Sketch

Geometric Properties

See log file listed bellow to see the coordinates of the contouring keypoints.

Materials

Layer 1

Organic High (OH) from CivilFEM library

Layer 2

Clay High (CH) from CivilFEM library

Layer 3

Sand Clay (SC) from CivilFEM library

Layer 4

Peat High (PH) from CivilFEM library

Analysis, Assumptions and Modeling Notes

Meters and Newton are used as length and stress units. Use PLANE 82 elements to mesh this model.

.

Input Data Listing

```
 FINISH
~CFCLEAR,,1! Not needed

~CFACTIV,GETC,Y   ! Activate geotechnical module (if available)

/TITLE,Slope estability analysis

! CivilFEM Setup
 ~UNITS,SI

/PREP7
! CivilFEM Preprocessor
! ---------------------------------------------------
! Materials
 ~CFMP,1,LIB,SOIL,,OH,0,0,0   ! Layer 1

 ~CFMP,2,LIB,SOIL,,CH,0,0,0   ! Layer 2

 ~CFMP,3,LIB,SOIL,,SC,0,0,0   ! Layer 3
 ~CFMP,3,SOIL,RUSI,,1,0,0,0   ! Activate susceptibility to pore pressure

 ~CFMP,4,LIB,SOIL,,PH,0,0,0   ! Layer 4

! Ansys Preprocessor
! ---------------------------------------------------
! Element type 1: Solid-2D
  ET,1,PLANE82

! Definition of contouring keypoints
 K, 1,-12,28
 K, 2,  1,29
 K, 3, 10,37
 K, 4, 19,38
 K, 5, 35,28
 K, 6, 50,27
 K, 7,  6,33
 K, 8, 20,29
 K, 9, 26,34
 K,10, 18,24
 K,11, 50,21
 K,12,-12,19
 K,13, 50,16
 K,14,-12, 6
 K,15, 50,10

! Generate areas (layers)
 A,7,8,9,4,3
 A,2,7,8,9,5,6,11,10
 A,1,2,10,11,13,12
 A,15,13,12,14

! Meshing
 AESIZE,ALL,3   ! Element size = 3m

! Mesh layers
 *DO,I,1,4
  MAT,I
  AMESH,I
 *ENDDO

! Apply concentrated loads (it must be applied on nodes!)
 P9=node(kx(9),ky(9),kz(9))  ! Node coincident with keypoint 9

 F,P9,FX,-94.86E4
 F,P9,FY,-31.622E4
! CAPTURING ANSYS MODEL
```

```
! --------------------------------------------------
! Nodes that define the model contour
 pp1=node(kx(1),ky(1),kz(1))
 pp2=node(kx(6),ky(6),kz(6))
 pp3=node(kx(5),ky(5),kz(5))

 ~SLPIN,PP1,PP2,PP3    ! Capture model

! Introduction of points defining the grid of centres
! --------------------------------------------------
 K,16,21,41
 K,17,35,32
 K,18,42,43
 K,19,28,52

~SLPCIRK,16,19,17,5,6

! Introduction of points defining the tangential lines
! --------------------------------------------------
 K,20,34,27
 K,21,34,11
 K,22, 0,11
 K,23,18,37

~SLPTANK,20,23,22,21,10

! Hydrostatic Pressure Definition
! --------------------------------------------------
~SLPPWP,5,0,-12,27, 01,27, 16,35, 41,25, 50,25

/POST1
! Solving slope
! --------------------------------------------------
 ~SLPSOL,1,0.3  ! Slides towards right and seismic horizontal coef. of 0.3

! Plot results in CivilFEM
! Visualize results
! CivilFEM Postprocessor > Geotech. Module > Slope stability > Results
```

CivilFEM Results



Figure 17.3-2 Minimum safety factor according to Bishop



Figure 17.3-3 Minimum safety factor according to Fellenius

Example 17.4
Slope Stability Analysis using F.E.M. Results

Example Description

Calculate the minimum safety factor and locate the critical slip surface for the same geometric model of the previous example using the results of a finite element method calculation.

In this case the model is subjected to a pressure applied on the top of the slope of 10KN/m and a concentrated load is applied on the middle of the slope. The piezometric line depicts the pore-water pressure conditions.

Special Features: Finite element method results, circular slip surfaces, 30 points in search grid, multiple soil layers, pore-water pressure specified by a piezometric line.





Figure 17.4-1 Problem Sketch

Geometric Properties

See log file listed bellow to see the coordinates of the contouring keypoints.

Materials

Layer 1

Organic High (OH) from CivilFEM library

Layer 2

Clay High (CH) from CivilFEM library

Layer 3

Sand Clay (SC) from CivilFEM library

Layer 4

Peat High (PH) from CivilFEM library

Analysis, Assumptions and Modeling Notes

Meters and Newton are used as length and stress units. Use PLANE 82 elements to mesh this model.

.

Input Data Listing

```
 FINISH
~CFCLEAR,,1! Not needed

~CFACTIV,GETC,Y   ! Activate geotechnical module (if available)

/TITLE,Slope estability analysis using FEM results

! CivilFEM Setup
 ~UNITS,SI

/PREP7
! CivilFEM Preprocessor
! --------------------------------------------------
! Materials
 ~CFMP,1,LIB,SOIL,,OH,0,0,0   ! Layer 1

 ~CFMP,2,LIB,SOIL,,CH,0,0,0   ! Layer 2

 ~CFMP,3,LIB,SOIL,,SC,0,0,0   ! Layer 3
 ~CFMP,3,SOIL,RUSI,,1,0,0,0   ! Activate susceptibility to pore pressure

 ~CFMP,4,LIB,SOIL,,PH,0,0,0   ! Layer 4

! Ansys Preprocessor
! --------------------------------------------------
! Element type 1: Solid-2D
  ET,1,PLANE82

! Definition of contouring keypoints
 K, 1,-12,28
 K, 2,  1,29
 K, 3, 10,37
 K, 4, 19,38
 K, 5, 35,28
 K, 6, 50,27
 K, 7,  6,33
 K, 8, 20,29
 K, 9, 26,34
 K,10, 18,24
 K,11, 50,21
 K,12,-12,19
 K,13, 50,16
 K,14,-12, 6
 K,15, 50,10

! Generate areas (layers)
 A,7,8,9,4,3
 A,2,7,8,9,5,6,11,10
 A,1,2,10,11,13,12
 A,15,13,12,14

! Meshing
 AESIZE,ALL,3   ! Element size = 3m
! Mesh layers
 *DO,I,1,4
  MAT,I
  AMESH,I
 *ENDDO

! Apply pressure
 KSEL,S,,,3,4
 LSLK,S,1
 NSLL,S,1
 SF,ALL,PRES,10000
 ALLSEL,ALL
! Apply concentrated loads (it must be applied on nodes!)
```

```
 P9=node(kx(9),ky(9),kz(9))  ! Node coincident with keypoint 9

 F,P9,FX,-94.86E4
 F,P9,FY,-31.622E4


! CAPTURING ANSYS MODEL
! --------------------------------------------------
! Nodes that define the model contour
 pp1=node(kx(1),ky(1),kz(1))
 pp2=node(kx(6),ky(6),kz(6))
 pp3=node(kx(5),ky(5),kz(5))

 ~SLPIN,PP1,PP2,PP3    ! Capture geometric model

! Introduction of points defining the grid of centres
! --------------------------------------------------
 K,16,21,41
 K,17,35,32
 K,18,42,43
 K,19,28,52

~SLPCIRK,16,19,17,5,6

! Introduction of points defining the tangential lines
! --------------------------------------------------
 K,20,34,27
 K,21,34,11
 K,22, 0,11
 K,23,18,37

~SLPTANK,20,23,22,21,10

! Hydrostatic Pressure Definition
! --------------------------------------------------
~SLPPWP,5,0,-12,27, 01,27, 16,35, 41,25, 50,25

/SOLU
! --------------------------------------------------
! Restraint conditions
 KSEL,S,,,14,15
 LSLK,S,1
 NSLL,S
 D,ALL,ALL,0
 ALLS
! Gravity
 ACEL,,9.8
! Solving the finite element model
 SOLVE

/POST1
! Solving slope
! --------------------------------------------------
 ~SLPOPT,,,,,2     ! Use finite element method results
 ~SLPSOL,10         ! Slides towards right

! Plot results in CivilFEM
! Visualize results
! CivilFEM Postprocessor > Geotech. Module > Slope stability > Results
```

CivilFEM Results



Figure 17.4-2 Geometric model



Figure 17.4-2 Minimum safety factor using FEM results

The black zone indicates that the program was not able to calculate the safety factor for this center (the slip circle for this center was out of the model).

Example 17.5
Slope Stability Analysis with Seepage

Example Description

Calculate the minimum safety factor and locate the critical slip surface, according to Janbu's method, for the model described in the figure bellow taking into account the influence of the seepage throughout it.



Figure 17.5-1 Problem Sketch

Geometric Properties

Keypoint 1 (0, 0)
Keypoint 2 (800, 0)
Keypoint 3 (0, 150)
Keypoint 4 (80, 150)
Keypoint 5 (232, 150)
Keypoint 6 (312, 150)
Keypoint 7 (520, 150)
Keypoint 8 (800, 150)
Keypoint 9 (223.3, 210)
Keypoint 10 (500.6, 160)
Keypoint 11 (295, 240)
Keypoint 12 (327.3, 240)
Keypoint 13 (345.5, 240)
Keypoint 14 (316.2, 160)

Initial saturation line coordinates

Initial point (X1): 295.5

Initial point (Y1): 210

End point (X2): 325

End point (Y2): 185

Number of points dividing the saturation line: 5

Line over which the initial point rest: line number 8

Line over which the second point rest: line number 13

Analysis, Assumptions and Modeling Notes

PLANE 42 elements are used to mesh this model (the program automatically meshes the model during the solution process).

Input Data Listing (Seepage Analysis)

```
  FINISH
 ~CFCLEAR,,1

 /FILNAME,wttoslp,0      ! Jobname

 ~CFACTIV,GETC,Y  ! Activate geotechnical module (if available)
 ~SEEPAGE,2D

 /TITLE, Slope stability with seepage
! --------------------------------------------------------------------------------
! Model definition
! --------------------------------------------------------------------------------
/PREP7
! CivilFEM Preprocessor
! --------------------------------------------------
! Materials: Soil

 ~CFMP,1,LIB,SOIL,,CL,0,0,0
 ~CFMP,2,LIB,SOIL,,GW,0,0,0
 ~CFMP,3,LIB,SOIL,,ML,0,0,0

! Ansys Preprocessor
! --------------------------------------------------
 PI = ACOS (-1)

! Define Keypoints
 K
 K,2,800,0,0
 K,3,0,150,0
 K,4,80,150,0
 K,5,232,150,0
 K,6,312,150,0
 K,7,520,150,0
 K,8,800,150,0
 K,9,223.3,210,0
 K,10,500.6,160
 K,11,295,240
 K,12,327.3,240
 K,13,345.5,240
 K,14,315.7,160

! Lines
 L,1,2
 L,3,4
 L,4,9
 L,9,11
 L,11,12
 L,13,10
 L,10,7
 L,5,12
 L,4,5
 L,5,6
 L,12,13
 L,6,14
 L,14,13
```

```
  L,6,7
  L,7,8
  L,1,3
  L,2,8
  L,10,14

! Define Water Table
  X1      = 295.5     ! X initial point
  Y1      = 210       ! Y initial point
  Z1      =   0       ! Z initial point
  X2      = 325       ! X end point
  Y2      = 185       ! Y end point
  Z2      =   0       ! Z end point
  NPUNTOS= 5          ! Number of points dividing line
  Linea1 = 8          ! Line over first point rest
  Linea2 = 13         ! Line over second point rest
  MAT     = 1         ! Material

 ~WATTAB,X1,Y1,Z1,X2,Y2,Z2,MAT,NPUNTOS,Linea1,Linea2,FIXED,EXIT


  L,9,15
  *GET,ULT_LIN,LINE,0,NUM,MAX

! Finite element model from lines
  MAT    = 3     !Material to be used
  ESIZE =  20    !Element size
  L1     = 1    $  L2     = 16  $   L3     = 2    $   L4     = 9
  L5     = 10   $  L6     = 14  $   L7     = 15   $   L8    = 17

 ~SEEPMOD,ADD,MAT ,,ESIZE,L1,L2,L3,L4,L5,L6,L7,L8    ! Area 1

  MAT    = 2
  ESIZE = 10
  L1     = 3    $  L2     = ULT_LIN  $   L3     = 8    $   L4     = 9

 ~SEEPMOD,ADD,MAT ,,ESIZE,L1,L2,L3,L4                 ! Area 2

  MAT    = 1
  ESIZE = 7
  L1     = 8    $  L2     = 10  $   L3     = 12   $   L4     = 13
  L5     = 24   $  L6     = 23  $   L7     = 22   $   L8     = 21
  L9     = 20   $  L10    = 19

 ~SEEPMOD,ADD,MAT ,WT,ESIZE,L1,L2,L3,L4,L5,L6,L7,L8,L9,L10  ! Area 3

  MAT    = 2
  ESIZE = 10
  L1     = 12    $  L2     = 18  $   L3     = 14    $   L4     = 7

 ~SEEPMOD,ADD,MAT ,,ESIZE,L1,L2,L3,L4                ! Area 4

/SOLU
! Ansys Solution
! --------------------------------------------------
! Generate boundary conditions (constant hydraulic head)
   HEAD1 = 210
   HEAD2 = 160
  ~DLHEAD,ADD,9,  HEAD1
  ~DLHEAD,ADD,3,  HEAD1
  ~DLHEAD,ADD,8,  HEAD1
  ~DLHEAD,ADD,2,  HEAD1
  ~DLHEAD,ADD,ULT_LIN,HEAD1
  ~DLHEAD,ADD,12,HEAD2
  ~DLHEAD,ADD,14,HEAD2
  ~DLHEAD,ADD,7 ,HEAD2
  ~DLHEAD,ADD,15,HEAD2
  ~DLHEAD,ADD,18,HEAD2
```

```
  ~DLSEEP,ADD,13

! Optimisation solver
  ~WTSOLVE
! Plot results
 ~ISOBAR
! Transfer the seepage results to the slope stability calculation
  ~WTSLP
```

CivilFEM Results



Figure 17.5-2 Isobar contour results

Input Data Listing (Slope stability using seepage results)

```
   FINISH
  ~CFCLEAR,,1

  /FILNAME,wttoslp,0     !Jobname

  /TITLE, Slope stability with seepage

/PREP7
! CivilFEM Preprocessor (We must redefine again the model)
! --------------------------------------------------
! Materials: Soil

  ~CFMP,1,LIB,SOIL,,CL,0,0,0
  ~CFMP,2,LIB,SOIL,,GW,0,0,0
  ~CFMP,3,LIB,SOIL,,ML,0,0,0
```

```
  ~CFMP,1,SOIL,RUSI,,1,0,0,0   ! Activate "rusi" parameter
  ~CFMP,2,SOIL,RUSI,,1,0,0,0   ! Activate "rusi" parameter
  ~CFMP,3,SOIL,RUSI,,1,0,0,0   ! Activate "rusi" parameter

! Ansys Preprocessor
! --------------------------------------------------
  PI = ACOS(-1)

! Keypoints
  K
  K,2,800,0,0
  K,3,0,150,0
  K,4,80,150,0
  K,5,232,150,0
  K,6,312,150,0
  K,7,520,150,0
  K,8,800,150,0
  K,9,223.3,210,0
  K,10,500.6,160
  K,11,295,240
  K,12,327.3,240
  K,13,345.5,240
  K,14,315.7,160

! Lines
  L,1,2
  L,3,4
  L,4,9
  L,9,11
  L,11,12
  L,13,10
  L,10,7
  L,5,12
  L,4,5
  L,5,6
  L,12,13
  L,6,14
  L,14,13
  L,6,7
  L,7,8
  L,1,3
  L,2,8
  L,10,14

! Meshing
  ESIZE,50
  AL,1,17,15,14,10,9,2,16 ! Base
  AL,9,8,5,4,3            ! Upstream face
  AL,14,7,6,13,12         ! Downstream face
  AL,11,8,10,12,13        ! Nucleus

  ET,1,200,6     ! Element type: MESH200
  TYPE,1
  MAT,3
  AMESH,1
  MAT,2
  AMESH,2
  AMESH,3
  MAT,1
  AMESH,4

! Nodes that define the model contour
  p1=NODE(KX(8),KY(8),KZ(8))
  p2=NODE(KX(3),KY(3),KZ(3))
  p3=NODE(KX(4),KY(4),KZ(4))

 ! SLOPE INPUT
  ~SLPIN,p1,p2,p3      ! Capture model
```

```
  ~SLPOPT,,,,,,3        ! Take thermal data into account

! Define Janbu polygonal
! ------------------------------------------------------------------
 ~SLPPOL, 200,220,200,150,100,75,45
 ~SLPPOL, 250,220,200,150,100,75,45
 ~SLPPOL, 300,220,200,150,100,75,45
 ~SLPPOL, 350,220,200,150,100,75,45
 ~SLPPOL, 400,220,200,150,100,75,45
 ~SLPPOL, 450,220,200,150,100,75,45
 ~SLPPOL, 500,220,200,150,100,75,45
 ~SLPPOL, 550,220,200,150,100,75,45
 ~SLPPOL, 600,220,200,150,100,75,45

/POST1
! --------------------------------------------------------
  ~SLPSOL,2     ! Solve the model

! Visualize results
! CivilFEM Postprocessor > Geotech. Module > Slope stability > Results
```



Figure 17.5-3 Sling surface according to Janbu

Example 17.6
Retaining Wall Calculation

Example Description

Determine the displacements and forces and moments law due to the loads and earth pressures acting over the two retaining walls of 30 m high linked by a joint as shown in the figure bellow. Afterwards check and design both of them against axial+bending according to the concrete Spanish code.



Figure 17.6-1 Problem Sketch

Materials

Soil	**Thickness for left wall**	**Thickness for right wall**
Organic Low	0,5	0,5
Well graduated gravel	7,0	4,0
Silt	3,5	3,5
Peat (Low)	40,0	43,0

Geometric properties

Retaining wall cross section:

Reinforcement: RKEY = 2



Anchor cross section:

Diameter: 0.3 m   Wall thickness: 0.03 m



Joint cross section:

HE 100 A

Analysis, Assumptions and Modeling Notes

Meters and Newton are used as length and stress units. Use BEAM 3 elements to mesh the joint, LINK 1 element to mesh the anchorages and BEAM 54 elements to mesh the walls.

Input Data Listing

```
   FINISH
  ~CFCLEAR,,1

  ~CFACTIV,GETC,Y  !Activate geotechnical module
! Initial data
! ---------------------------------------------------------
! CivilFEM SETUP
  ~UNITS,SI
   ~CODESEL,EC3-05,EC2-08,,,EC8-04

/PREP7
! ELEMENT TYPES
! --------------------------------------------------
 ET,1,BEAM3
 ET,2,LINK1
 ET,10,BEAM54

! MATERIALS
```

```
! ---------------------------------------------------
! Steel
~CFMP,1,LIB,STEEL,UNE,S 235,0,0,0
! Reinforced concrete
~CFMP,12,LIB,CONCRETE,EC2,C16/20,0,0,0
~CFMP,23,LIB,REINF,EC2,S400,0,0,0
! Soils and rocks
~CFMP,101,LIB,SOIL,,OL
~CFMP,102,LIB,SOIL,,GW
~CFMP,103,LIB,SOIL,,ML
~CFMP,104,LIB,SOIL,,PL

! CROSS SECTIONS
! ---------------------------------------------------
! Joint
~SSECLIB,1,1,5,1
~SECMDF,1,NAME,,, ! Joint

! Anchor
~SSECDMS,2,PIPE,1,0.3,3e-002,0,0,,0
~SECMDF,2,NAME,,, ! Anchor

! Wall
~CSECDMS,3,REC,12,1,10,0,0,0,0,0,0,2,23
~SECMDF,3,NAME,,, ! Wall Section

! BEAM & SHELL PROPERTIES
! ---------------------------------------------------
! Joint
~BMSHPRO, 1,BEAM,1,1,,,3,1,0, ,Joint

! Anchor
~BMSHPRO, 2,BEAM,2,2,,,1,1,0, , Anchor

! Wall
~BMSHPRO, 3,BEAM,3,3,,,54,1,0, , Wall section

! TERRAINS
! ---------------------------------------------------
! ~TERDEF,UETH,NEW,NLAYER,EAKCN,EADIR,SURFLV,BETA,Q,WKEY,WH,NAME
  ~TERDEF,   1,NEW,     4,    0,     Y,  1045,    0,1000
! ~TERDEF,UETH,LAYER,ILAYER,MAT,THK, HBM,KSEL
  ~TERDEF,   1,LAYER,     1,101,0.5,    0,  DB
  ~TERDEF,   1,LAYER,     2,102,  7,    0,  DB
  ~TERDEF,   1,LAYER,     3,103,3.5,    0,  DB
  ~TERDEF,   1,LAYER,     4,104, 40,    0,CLMB

! ~TERDEF,UETH,NEW,NLAYER,EAKCN,EADIR,SURFLV,BETA,Q,WKEY,WH,NAME
  ~TERDEF,   2,NEW,     4,    0,     Y,  1045
! ~TERDEF,UETH,LAYER,ILAYER,MAT,THK, HBM,KSEL
  ~TERDEF,   2,LAYER,     1,101,0.5,    0,  DB
  ~TERDEF,   2,LAYER,     2,102,  4,    0,  DB
  ~TERDEF,   2,LAYER,     3,103,3.5,    0,  DB
  ~TERDEF,   2,LAYER,     4,104, 40,    0,CLMB

! WALL
! Initialization
  ~WALLINI,   0,   20,    0,    0,    1,    1, 1046, 10,   20,    2,    1, 1046, 10

! Wall definition
  ~WALLGEN,    1,    1,    30, 12,   10,   3
  ~WALLGEN,    2,    1,    30, 12,   10,   3

! Excavation
  *DO,STEP,1,20
    ~WALLSTP,STEP,    1,1045-STEP
    ~WALLSTP,STEP,    2,1045-STEP
  *ENDDO
```

```
! Anchorages
  ~WALLANC,    9,     1,    1037,       1,   170,      5,   1,    2,    2
  ~WALLANC,    9,     2,    1037,       1,    10,      5,   1,    2,    2

! Joint between walls
  ~WALLJNT,    5,     1043,     1043,   1,    1,    1

! Create model. Maximum element size = 0.5 m
  ~WALLMOD, 0.5

 /PBC,ALL, ,1
 /SHRINK,0.1
 eplot

/SOLU
! SOLUTION
! -------------------------------------------------------
 neqit,150

~WALLSOL, ALL

/POST1
! -------------------------------------------------------
! Read results
~CFSET,0,20,LAST,    !Last load step (20)
!Plot the deformed shape
PLNSOL,U,X,0,1
!Plot the forces and moments on the retaining wall
 ESEL,S,TYPE,,10     ! Select retaining wall elements (Beam 54)
~PLLSFOR,M,Z,-1,     ! Bending moment Z
~PLLSFOR,F,X,-1,     ! Axial force
 ESEL,ALL
! Check the retaining wall for axial+bending
~CHKCON,2DB, , ,0
~PLLSCON,CRT_TOT,1,
! Design the retaining wall for axial+bending
~DIMCON,2DB, ,0,0.5,10,
 ESEL,S,TYPE,,10     ! Select retaining wall elements (Beam 54)
~PLLSCON,REINFACT,1,
 ESEL,ALL
```

CivilFEM Results



Figure 17.6-2 Finite element model



Figure 17.6-3 Displacement at X direction



Figure 17.6-3 Bending moment distribution



Figure 17.6-4 Criteria for axial+bending checking



Figure 17.6-5 Reinforcement factor for the retaining wall

Example 17.7
Hoek & Brown Failure Criteria

Example Description

Determine the stresses and displacements for the structure shown in the figure bellow. The structure is in an infinite elastic-plastic medium subjected to a constant in-situ compressive stress field of 30 MPa (that should be introduced as an initial stress state in ANSYS). The material is assumed to be linearly elastic and perfectly plastic with a failure surface following the Hoek & Brown failure criteria. The geometry of the structure as well as the material properties are defined here in.



Figure 17.7-1 Problem Sketch

Material Properties

Rock from CivilFEM library: Amphibol

Elasticity modulus (E) = 10000 Mpa

Compressive strength = 150 MPa

Hoek & Brown coefficients of the intact rock (the coefficients of the fractured rock are not considered in this analysis):

m = 2.94

s = 0.01

Analysis, Assumptions and Modeling Notes

Meters and Kilonewton are used as length and stress units. The radius of the hole is small compared to the length of the cylinder, therefore 2D plane strain are in effect. Use PLANE 42 elements.

Input Data Listing

```
  FINISH
~CFCLEAR,,1

/TITLE, Cylindrical Hole in an Infinite Hoek - Brown Medium
! Units
! -----------------------------------------------------------------
 ~UNITS,,LENG,M
 ~UNITS,,TIME,S
 ~UNITS,,FORC,KN

 /PREP7
 ANTYPE,STATIC          ! STATIC ANALYSIS
! CivilFEM Preprocessor
! --------------------------------------------------
! Material: ROCK
    ~CFMP,1,LIB,ROCK,,Amphibol,0,0,0
    ~CFMP,1,ROCK,GSI ,,30,0,0,0       ! GSI = 30
    ~CFMP,1,ROCK,HB_M,,2.94,0,0,0     ! m = 2.94
    ~CFMP,1,ROCK,HB_S,,1e-002,0,0,0   ! s= 0.01
    ~CFMP,1,ROCK,QU  ,,150e3,0,0      ! Compressive strength = 150 MPa
    ~CFMP,1,ROCK,EXST,,10000e3,0,0,0 ! E   = 10000 MPa
    ~CFMP,1,ROCK,KPLA,,1,0,0,0

! ANSYS PREPROCESSOR
! Element type
! -------------------------------------------
    ET,1,PLANE42,,,2         ! Plane strain
! Solid model
    ri = 10   ! Internal radius (m)
    re = 100 ! External radius
  ! Load
    Po = 30e3  ! Initial stress MPa
    msiz = ri  ! Element division
  ! Keypoints & Lines
    K,100
    K,1,RI
    K,2,,RI
    LARC,1,2,100,RI
    K,3,100
    K,4,,100
    LARC,3,4,100,100
    L,3,1
    L,4,2
    AL,ALL

! Finite element Model
! --------------------------------------------------
    LESIZE,3,,,msiz*2,0.005  !Element division per line
    LESIZE,1,,,msiz*2
    LESIZE,4,,,msiz*2,0.005
    LESIZE,2,,,msiz*2
    AMESH,ALL

    ARSYM,X,ALL    !Generate area by symmetry on X
    ARSYM,Y,ALL    !Generate area by symmetry on Y
```

```
    NUMMRG,NODE
```

```
    NUMMRG,KP
    EPLOT

/SOLU
! Solution Options
! --------------------------------------------------
    AUTOTS,ON
   !NSUBST,10,2000,5
    NEQIT,150

! Boundary Conditions
! --------------------------------------------------
    NSEL,S,LOC,Y,0
    NSEL,R,LOC,X,RI
    D,ALL,UY,0
    NSEL,S,LOC,Y,0
    NSEL,R,LOC,X,-RI
    D,ALL,UY,0
    NSEL,ALL
    NSEL,S,LOC,X,0
    NSEL,R,LOC,Y,RI
    D,ALL,UX,0
    NSEL,S,LOC,X,0
    NSEL,R,LOC,Y,-RI
    D,ALL,UX,0
    ALLS

! Loads
! --------------------------------------------------
    SFL,2 ,PRES,Po    !Pressure on lines
    SFL,7 ,PRES,Po
    SFL,15,PRES,Po
    SFL,11,PRES,Po

! Define initial stress state
! --------------------------------------------------
    ISTRESS,-Po,-Po,,,,,1

! Configuration parameters
! --------------------------------------------------
  ~CFCONFG,HB,HBSOL,1    ! Solve by substeps
  ~CFCONFG,HB,RESIDU,0   ! No residual strength

! Solve the model

  NSUBST,1000,1000,10
  CUTCONTROL,PLSLIMIT,1.
  ~HBSOLVE              ! Solve the model following Hoek & Brown criteria

/POST1
! CivilFEM POSTPROCESSOR
! --------------------------------------------------
! Read results
 ~CFSET,,1
! Plot displacements
 PLNSOL,U,X,0,1       ! Displacements at X direction
 PLNSOL,U,Y,0,1       ! Displacements at Y direction
! Plot stresses
 PLNSOL,S,X,0,1
 PLNSOL,S,Y,0,1
! Plot friction angle variation
 ~PLHBMAT,PHI
! Plot cohesion variation
 ~PLHBMAT,COH
```

CivilFEM Results



Figure 17.7-2 Displacements at X direction



Figure 17.7-3 Displacements at Y direction



Figure 17.7-4 Stress at Y direction



Figure 17.7-5 Friction angle contour